# $Id: PKGBUILD 71776 2012-06-02 10:30:20Z bluewind $
# Maintainer: Sébastien Luttringer <seblu@aur.archlinux.org>

pkgname=perl-file-rsyncp
pkgver=0.70
pkgrel=2
pkgdesc='Perl interface to rsync file list encoding and decoding'
arch=('i686' 'x86_64')
license=('GPL' 'PerlArtistic')
options=('!emptydirs' '!makeflags')
url='http://search.cpan.org/dist/File-RsyncP'
source=("http://search.cpan.org/CPAN/authors/id/C/CB/CBARRATT/File-RsyncP-$pkgver.tar.gz")
md5sums=('f244372d15a2991b8700f62e73ac51e4')

build() {
  cd File-RsyncP-$pkgver
  USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd File-RsyncP-$pkgver
  make test
}

package() {
  cd File-RsyncP-$pkgver
  make install DESTDIR="$pkgdir/"
}

# vim:set ts=2 sw=2 ft=sh et:
